*==========================================================================================================================================================
*
*	Local Projections
*
*==========================================================================================================================================================


set more off

qui {


*-------------------------------------------------------------------------------------------------------------
*	SETTINGS
*-------------------------------------------------------------------------------------------------------------

* Variables
local vars "p_beef_index p_mutton_index" 

* irf horizon
local H 5
local F = 5

* lags
local L 2 
local LS `L' 

*-------------------------------------------------------------------------------------------------------------



*_________ GET DATA ____________
use "data/working files/prepared.dta", clear 


*____ DECLARE TIME SERIES _____
gen id =1
tsset id year, yearly


*_________________ LOCAL PROJECTIONS _________________

local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

***_________ CONTROLS

local indep	" gdp_r_adj_gr wage_sil_gr itsN_gr" 

forvalues i=0/`L'{
	local indep_`i' = subinstr("`indep'"," "," l`i'_",.)
	}
	
local rhs ""
forvalues i = 1/`L'{
	local rhs "`rhs'`indep_`i''"
	}
	
	local not l0_`v'_gr l0_itsN_gr
	local rhs: list rhs- not
	

	***_________ IMPULSE RESPONSE and CONFIDENCE INTERVAL variables
	gen irf_`v' = . if _n<=`H'+1
	
	gen se_`v' = . if _n<=`H'+1

	gen sig_`v' = . if _n<=`H'+1
	gen ci_up_`v' = . if _n<=`H'+1
	gen ci_lo_`v' = . if _n<=`H'+1
	gen ci_up90_`v' = . if _n<=`H'+1
	gen ci_lo90_`v' = . if _n<=`H'+1
				
	
	***_________ ESTIMATE COEFFICIENTS
	
	forvalues y = 0/`H' {
		n display "projection `y'"

		newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').perclossN_vellon l(-`F'/`LS').perclossN l(-`F'/`LS').percloss_salvN `rhs' l(-`F'/`LS').percloss_pirN l(0/`LS').temperature l(0/`LS').warbeu l(0/`LS').warbciv, lag(5) force	
	
		* store IMPULSE RESPONSES and STANDARD ERRORS	
		lincom perclossN 
		replace irf_`v' = r(estimate) if _n==`y'+1 /* coefficients */
		replace se_`v' = r(se) if _n==`y'+1 /* standard errors */
		

		} //horizon			
	} // vars

	
local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

	***_________ CONFIDENCE INTERVALS
	replace ci_up_`v' = irf_`v' + 1*se_`v' if _n <= `H'+1
	replace ci_lo_`v' = irf_`v' - 1*se_`v' if _n <= `H'+1
	replace ci_up90_`v' = irf_`v' + 1.64*se_`v' if _n <= `H'+1
	replace ci_lo90_`v' = irf_`v' - 1.64*se_`v' if _n <= `H'+1
	
	replace sig_`v' =  irf_`v' if ci_up90_`v' < 0 | ci_lo90_`v' > 0

} // vars

* plot prerequisites
gen time = _n-1 if(_n<=`H'+1) 
drop if time > `H'
keep time irf_* ci_up_* ci_lo_*  ci_up90_* ci_lo90_* se_* sig_*
order irf_* ci_up_* ci_lo_*  time ci_up90_* ci_lo90_* sig_*
drop if time == . 
gen zero = 0 if time!=.
gen time0 = time
gen time1 = time -0.025
gen time2 = time + 0.025

local lsize "medium"
local tsize "medium"
local legsize "medium"

local zline "medium"
local irfline "medthick"
local msize "medlarge"


} // qui

//Primary: meat

twoway  (line zero time, lcolor(black) lwidth(`zline')) ///
		(scatter irf_p_mutton_index time, color(black) msymbol(circle_hollow)  msize(`msize')) ///
		(line irf_p_mutton_index time, lcolor(black) lpattern(dot) lwidth(`irfline')) ///
		(scatter irf_p_beef_index time, color(gs8) msymbol(diamond_hollow)  msize(`msize')) ///
		(line irf_p_beef_index time, lcolor(gs8) lpattern(dot) lwidth(`irfline')) ///
		(scatter sig_p_mutton_index time, color(black) msymbol(circle)  msize(`msize')) ///
		(scatter sig_p_beef_index time, color(gs8) msymbol(diamond)  msize(`msize')), ///
		title("Meats", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(-4(1)2, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4) col(1) pos(7) ring(0) lab(2 "Mutton") lab(4 "Beef") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g1, replace) nodraw
		
		


		
graph combine g1, name(LP_meat, replace) cols(1) imargin(medlarge) plotregion(color(white)) graphregion(color(white)) 
graph display LP_meat, ysize(2.1) xsize(2.5) scale(1)
graph export "results/FigureB2.pdf", replace 


	

*_________________________________________

			

			
